

# Реализация виджета средствами QML

## Общие сведения

Данный материал является приложением к мастер-классу Евгения Горелова [«Разработка элементов графического интерфейса в среде Qt»](#) и содержит подробное описание QML кода.

## Описание кода

- 1) Для создания нового элемента объявляется тип *Item*, задаются исходные значения ширины и высоты. Создаваемому элементу добавляются целочисленные свойства минимального и максимального значения угла (*min* и *max*), а также текущее значение угла наклона стрелки (*rAngle*).

```
Item {
    width: 100
    height: 62

    property int min: -59
    property int max: 58

    property real rAngle: 50
```

- 2) Внутри родительского (корневого элемента) создаётся элемент *Timer*, отвечающий за изменение параметра *rAngle* (угла наклона стрелки).

Таймеру задаётся интервал (*interval*), текущее состояние (*running*), повторение (*repeat*). В качестве дополнительного параметра задаётся приращение угла *cr*.

Значение *cr* (положительное/отрицательное) инвертируется каждый раз при достижении *rAngle* минимума и максимума. Вычисления производятся внутри обработчика срабатывания таймера *onTriggered*.

```
Timer{
    property int cr: 1

    interval: 200; running: true; repeat: true
    onTriggered:{

        rAngle += cr

        if( rAngle == min || rAngle == max )
        {
            cr = -cr
        }
    }
}
```

- 3) Корпус прибора состоит из двух изображений: *meter\_top.png* и *meter\_bottom.png*. Для их отображения используется элемент *Image*. Первому элементу задаётся уникальный идентификатор (*id*) *meterTop*, исходный файл изображения (*source*), а также якорь для относительного позиционирования.

```
Image{
    id: meterTop
    source: "images/meter_top.png"
    anchors.centerIn: parent
}
```

Второму элементу задаётся исходный файл изображения, а также якоря для относительного позиционирования. Элемент располагается относительно горизонтального центра родительского элемента (*parent.horizontalCenter*) и нижнего края элемента *meterTop* (*meterTop.bottom*). Уникальный идентификатор (*id*) второму элементу не задаётся, т.к. нет необходимости обращения к нему и его свойствам откуда-либо извне.

```
Image{
    id: meterBottom
    source: "images/meter_bottom.png"
    anchors.horizontalCenter: parent.horizontalCenter
    anchors.top: meterTop.bottom
}
```

- 4) *meterTop* содержит внутри себя три 3 элемента:

- a) Текст с обозначением единиц измерения.

```
Text{
    text: "V"
    font.pointSize: 14
    anchors.horizontalCenter: parent.horizontalCenter
    anchors.bottom: parent.bottom
    anchors.bottomMargin: 40
}
```

- b) Изображение стрелки.

Вращение (*Rotation*) можно задать с помощью атрибута *transform*. Элементу *Rotation* задаются ось и координаты точки вращения, а также угол вращения *angle* приравненный к *rAngle*. Для плавного изменения угла вращения определяется поведение *Behavior* при изменении угла *angle*. В данном случае используется анимация, которая постепенно изменяет параметр в течение заданного времени (300 мс). Для минимизации искажений при вращении, свойству *smooth* (отвечает за сглаживание) изображения *arrow* задано значение *true*.

```

Image{
    id: arrow
    source: "images/arrow.png"

    anchors.horizontalCenter: parent.horizontalCenter
    anchors.bottom: parent.bottom
    smooth: true

    transform: Rotation{
        id: arrowRotation
        axis.x: 0
        axis.y: 0
        axis.z: 1

        origin.x: arrow.width/2
        origin.y: arrow.height

        angle: rAngle

        Behavior on angle{ PropertyAnimation{ duration: 300 } }
    }
}

```

с) Изображение основания стрелки.

```

Image{
    id: arrowBase
    source: "images/arrow_base.png"
    anchors.horizontalCenter: parent.horizontalCenter
    anchors.bottom: parent.bottom
}

```

[Обратная связь](#)

СВД Встраиваемые Системы, <http://www.kpda.ru>

[Страница проекта](#)